

EIN TASCHENCOMPUTERPROGRAMM ZUR DIFFERENTIALDIAGNOSE VON TUMORINDUZIERTER HYPERCALCÄMIE UND PRIMÄREM HYPERPARATHYREOIDISMUS.

M. Ledořowski, M. Herold, F. Dienstl

Univ.-Klinik für Innere Medizin, Abteilung für Kardiologische Intensivmedizin der Universität Innsbruck.

Die rasche Unterscheidung zwischen tumorinduzierter Hypercalcämie und primärem Hyperparathyreoidismus stellt immer noch ein Problem dar, zumal die Bestimmung einzelner biochemischer Parameter wie PTH-Spiegel, ionisiertes Serumcalcium, TRCa, IPE, etc., meist lange dauern, nicht überall durchführbar oder ungenau sind. Dies ist vor allem bei der hypercalcämischen Krise von Bedeutung. Anhand einer von Watson et al erstellten Diskriminanzanalyse wurde für einen programmierbaren Taschenrechner (HP 71B) ein Programm in BASIC erstellt, mit dessen Hilfe mit 93 prozentiger Wahrscheinlichkeit die der Hypercalcämie zugrundeliegende Ursache bestimmt werden kann. Die dafür benötigten Daten sind BSG, Bicarbonat, Harnstoff, Alkalische Phosphatase, Phosphat und Chlorid im Serum, welche alle notfallsmäßig bestimmt werden können. Bei der Überprüfung des Programms anhand von vier Patienten konnte in allen Fällen eine Übereinstimmung der Computerdiagnose mit der klinischen bzw. histologischen Diagnose gefunden werden.

Wir glauben, daß mit diesem Programm eine verbesserte und beschleunigte Differentialdiagnostik der hypercalcämischen Krise möglich ist.

```

10 ! CALDIF- c by 1985 Herold/Ledochowski
20 DESTROY ALL @ M(5)=INF @ M(6)=INF @ INPUT "Calcium (mmol/l) ?";D
30 IF D>2.7 THEN 50
40 BEEP @ DISP "No hypercalcemia !" @ STOP @ GOTO 20
50 INPUT "ESR (mm l.value) ? ";S
60 INPUT "Urea (mmol/l) ?";U
70 INPUT "Chloride (mmol/l) ?";C
80 INPUT "Inorg.Phos.(mmol/l) ?";P
90 INPUT "Alk.Phos.(U/l-25) ? ";A
100 INPUT "Bic. (mmol/l) ? ";B
110 A=A*.61 ! U=U*.1666
120 X=4.0532+3.8375*P-.823*LGT(A)-.1095*C+.0048*B+4.2989*LGT(U)+.0125*S
130 Y=1.7088-.9198*P-3.6206*LGT(A)-.1030*C+.1475*B-1.7608*LGT(U)+.0025*S
140 M(0)=SQR((X+2.69)^2+(Y+13.48)^2)
150 M(1)=SQR((X+2.79)^2+(Y+17.33)^2)
160 M(2)=SQR((X-1.37)^2+(Y+14.45)^2)
170 M(3)=SQR((X-6.61)^2+(Y+14.86)^2)
180 M(4)=SQR((X-1.76)^2+(Y+19.64)^2)
190 FOR I=0 TO 4
200 IF M(I)<M(5) THEN M(5)=M(I)
210 NEXT I
220 FOR I=0 TO 4
230 IF M(I)=M(5) THEN 250
240 IF M(I)<M(6) THEN M(6)=M(I)
250 NEXT I
260 DELAY INF @ FIX 2 @ IF M(5)<=2.45 THEN 280
270 BEEP @ DISP "Imprecise assertion !"
280 DISP "best fit:";M(5);"sdu" @ GOSUB 320
290 IF M(6)<=2.45 THEN I=6 @ DISP "2nd best fit:";M(6);"sdu" @ GOSUB 320
300 DISP "End / Return "
310 IF KEYDOWN("R") THEN 260 ELSE 380
320 IF M(I)=M(0) THEN DISP "HPTH - (Area A)"
330 IF M(I)=M(1) THEN DISP "HPTH - (Area B)"
340 IF M(I)=M(2) THEN DISP "NON-HPTH - (Area C)"
350 IF M(I)=M(3) THEN DISP "NON-HPTH - (Area D)"
360 IF M(I)=M(4) THEN DISP "HPTH - (Area E)"
370 RETURN
380 DISP "END" @ END

```

Table 1:

Printlisting of the programme CALDIF as it runs on a HP-71B computer
If urea is to be entered in (mg/dl) the ! should be erased in line 110